>

LaPlace Equation

A. D. Andrew

Math 4581

27 February 2007

Example 1.

>

In this example we consider the LaPlace equation in the unit square, with boundary values 0 on the left and bottom edges, f(x) = x on the top, and g(y) = y on the right.

We split the original in two sub problems which are in fact the same. The first is with 0 boundary data on all sides save the top, and f(x) = x there. The second has 0 boundary data on all sides but the right, and g(y) = y on the right side.

If the solution to the first sub problem is u(x,y), then the solution to the original problem is v(x,y) = u(x,y) + u(y,x).

By the analysis done in class, a partial sum for u(x,y) is

> $u1 := (x, y, N) \rightarrow add\left(\frac{-2\cdot(-1)^n}{(n\cdot\pi\cdot\sinh(n\cdot\pi))}\cdot\sinh(n\cdot\pi\cdot y)\cdot\sin(n\cdot\pi\cdot x),\ n = 1..N\right);$

$$u1 := (x, y, N) \rightarrow add\left(-\frac{2\,(-1)^n \sinh(n\,\pi\,y)\,\sin(n\,\pi\,x)}{n\,\pi\,\sinh(n\,\pi)},\ n = 1..N\right) \quad (1.1)$$

>

so we take

> $v1 := (x, y, N) \rightarrow u1(x, y, N) + u1(y, x, N);$

$$v1 := (x, y, N) \rightarrow u1(x, y, N) + u1(y, x, N) \quad (1.2)$$

>

Here is a plot and contour plot for N = 10.

> $plot3d(v1(x, y, 10),\ x = 0..1,\ y = 0..1,\ axes = boxed,\ shading = none);$



> $with(plots):$

> $contourplot(v1(x, y, 20), x = 0..1, y = 0..1, color = black);$



>

The unusual appearance near the upper right occurs because we introduced discontinuities in boundary data when we separated the problem into two. This is the resulting Gibbs phenomenon.

>

Example 2

Here we put boundary conditions of 0 on the bottom and left, 1 on top, and ux = 0 on the right. The eigenvalues become

> $\lambda 2 := n \rightarrow (2 \cdot n + 1) \cdot \left(\frac{\pi}{2}\right);$

$$\lambda 2 := n \rightarrow \frac{1}{2} \left(2\, n + 1\right) \pi \tag{1.1.1}$$

> $\lambda 2(5);$

(1.1.2)

$$\frac{11}{2}\pi \qquad (1.1.2)$$

and a partial sum of the solution is

$$> u2 := (x, y, N) \rightarrow add\left(\frac{2}{\lambda 2(n)\cdot\sinh(\lambda 2(n))}\cdot\sinh(\lambda 2(n)\cdot y)\cdot\sin(\lambda 2(n)\cdot x),\ n = 0..N\right);$$

$$u2 := (x, y, N) \rightarrow add\left(\frac{2\sinh(\lambda 2(n)\, y)\,\sin(\lambda 2(n)\, x)}{\lambda 2(n)\,\sinh(\lambda 2(n))},\ n = 0..N\right) \qquad (1.1.3)$$

$$> evalf(u2(.5, .5, 20));$$

$$0.3640566635 \qquad (1.1.4)$$

A contour plot is

$$> contourplot(u2(x, y, 10), x = 0..1, y = 0..1, color = black);$$



> $plot3d(u2(x, y, 10), x = 0..1, y = 0..1, shading = none, axes = boxed);$



>

Example 3

This time we put 0 boundary values on top and bottom and f(y) = sin(Piy) on the left and right edges. This is the sum of two similar problems, the first of which has nonzero boundary values only on the right edge and the second on the left edge. If he solution to the first of these is u(x,y), then the solution to the second is u(1−x,y). The first problem has solution

> $u3 := (x, y) \rightarrow \frac{(\sin(\pi \cdot y) \cdot \sinh(\pi \cdot x))}{\sinh(\pi)};$

$$u3 := (x, y) \rightarrow \frac{\sin(\pi\, y)\, \sinh(\pi\, x)}{\sinh(\pi)} \qquad (1.2.1)$$

>

and the solution to the original problem is

> $v3 := (x, y) \rightarrow u3(x, y) + u3(1-x, y);$

$$v3 := (x, y) \rightarrow u3(x, y) + u3(1-x, y) \qquad (1.2.2)$$

> $plot3d(v3(x, y), x = 0..1, y = 0..1, shading = none, axes = boxed);$



>

>

Example 4

In polar coordinates, we consider LaPlace's equation on the disk, with boundary values 1 on the upper half disk and 0 on the lower. A partial sum for the solution, as derived in class, is

>

> $u4 := (r, t, N) \rightarrow piecewise\left(r < 1 \text{ and } r \geq 0, \frac{1}{2} + add\left(\frac{1}{n \cdot \pi} \cdot (1 - (-1)^n) \cdot r^n \cdot \sin(n \cdot t), n = 1..N\right), 1000\right);$

$$u4 := (r, t, N) \to piecewise\left(r < 1 \textbf{ and } 0 \le r, \frac{1}{2} + add\left(\frac{(1-(-1)^n)\, r^n \sin(n\,t)}{n\,\pi}, n = 1..N\right), 1000\right) \quad (1.3.1)$$

>

This formula is more complicated than necessary. The "piecewise" and "1000" were included to help trick Maple into sketching an attractive contour plot. I also found it necessary to use a cartesian coordinate formulation.

>

> $r := (x, y) \to \text{sqrt}(x^2 + y^2);$

$$r := (x, y) \to \sqrt{x^2 + y^2} \quad (1.3.2)$$

> $t := (x, y) \to \arctan(y, x);$

$$t := (x, y) \to \arctan(y, x) \quad (1.3.3)$$

Here's the contour plot.

> $with(plots):$

> $contourplot(u4(r(x, y), t(x, y), 100), x = -1..1, y = -1..1, contours = [0, .1, .2, .3, .4, .5, .6, .7, .8, .9, 1], grid = [150, 150], color = black);$



Yes, those are arcs of circles. Why?

>

>

>